SIMULACIÓN DE MONTE CARLO CON EXCEL

Autores: Javier Faulín (ffaulin@uoc.edu), Ángel A. Juan (ajuanp@uoc.edu).

ESQUEMA DE CONTENIDOS



INTRODUCCIÓN

La simulación de Monte Carlo es una técnica que combina conceptos estadísticos (muestreo aleatorio) con la capacidad que tienen los ordenadores para generar números pseudo-aleatorios y automatizar cálculos.

Los orígenes de esta técnica están ligados al trabajo desarrollado por Stan Ulam y John Von Neumann a finales de los 40 en el laboratorio de Los Alamos, cuando investigaban el movimiento aleatorio de los neutrones **[W1]**. En años posteriores, la simulación de Monte Carlo se ha venido aplicando a una infinidad de ámbitos como alternativa a los modelos matemáticos exactos o incluso como único medio de estimar soluciones para problemas complejos. Así, en la actualidad es posible encontrar modelos que hacen uso de simulación MC en las áreas informática, empresarial, económica, industrial e incluso social **[5, 8]**. En otras palabras, la simulación de Monte Carlo está presente en todos aquellos ámbitos en los que el comportamiento aleatorio o probabilístico desempeña un papel fundamental -precisamente, el nombre de Monte Carlo proviene de la famosa ciudad de Mónaco, donde abundan los casinos de juego y donde el azar, la probabilidad y el comportamiento aleatorio conforman todo un estilo de vida.

Son muchos los autores que han apostado por utilizar hojas de cálculo para realizar simulación MC **[1, 6, 7]**. La potencia de las hojas de cálculo reside en su universalidad, en su facilidad de uso, en su capacidad para recalcular valores y, sobre todo, en las posibilidades que ofrece con respecto al análisis de escenarios ("*what-if anaylisis*"). Las últimas versiones de Excel incorporan, además, un lenguaje de programación propio, el *Visual Basic for Applications*, con el cual es posible crear auténticas aplicaciones de simulación destinadas al usuario final. En el mercado existen de hecho varios complementos de Excel (*Add-Ins*) específicamente diseñados para realizar simulación MC, siendo los más conocidos: @Risk, Crystall Ball, Insight.xla, SimTools.xla, etc. **[W2 – W5]**.

OBJETIVOS

- Introducir los conceptos e ideas clave de la simulación MC.
- Introducirse en las capacidades que ofrece Excel en los campos de modelado y simulación.
- Conocer algunas aplicaciones de la simulación MC.

CONOCIMIENTOS PREVIOS

Este *math-block* supone ciertos conocimientos básicos de estadística (inferencia y probabilidad), así como conocimientos -a nivel de usuario- de la hoja de cálculo Excel.

CONCEPTOS FUNDAMENTALES

La función ALEATORIO() de Excel

Las hojas de cálculo como Excel (y cualquier lenguaje de programación estándar) son capaces de generar números pseudo-aleatorios provenientes de una distribución uniforme entre el 0 y el 1. Este tipo de números pseudo-aleatorios son los elementos básicos a partir de los cuales se desarrolla cualquier simulación por ordenador.

En Excel, es posible obtener un número pseudo-aleatorio -proveniente de una distribución uniforme entre el 0 y el 1- usando la función **ALEATORIO**:



Los números generados mediante la función ALEATORIO tienen dos propiedades que los hacen equiparables a números completamente aleatorios:

1. Cada vez que se usa la función ALEATORIO, cualquier número real entre 0 y 1 tiene la misma probabilidad de ser generado (de ahí el nombre de distribución uniforme).

2. Los diferentes números generados son estadísticamente independientes unos de otros (es decir, el valor del número generado en un momento dado no depende de los generados con anterioridad).

La función ALEATORIO es una función **volátil** de Excel. Esto significa que cada vez que pulsamos la tecla **F9** o cambiemos alguno de los *inputs* del modelo, todas las celdas donde aparezca la función ALEATORIO serán recalculadas de forma automática.

Se pueden encontrar ejemplos del uso de ALEATORIO en el propio menú de ayuda de Excel.

¿Qué es la simulación de Monte Carlo?

La **simulación de Monte Carlo** es una técnica cuantitativa que hace uso de la estadística y los ordenadores para imitar, mediante modelos matemáticos, el comportamiento aleatorio de sistemas reales no dinámicos (por lo general, cuando se trata de sistemas cuyo estado va cambiando con el paso del tiempo, se recurre bien a la simulación de eventos discretos o bien a la simulación de sistemas continuos).

La clave de la simulación MC consiste en crear un modelo matemático del sistema, proceso o actividad que se quiere analizar, identificando aquellas variables (*inputs* del modelo) cuyo comportamiento aleatorio determina el comportamiento global del sistema. Una vez identificados dichos *inputs* o variables aleatorias, se lleva a cabo un experimento consistente en (1) **generar** –con ayuda del ordenador- **muestras aleatorias** (valores concretos) para dichos *inputs,* y (2) analizar el comportamiento del sistema ante los valores generados. Tras repetir *n* veces este experimento, dispondremos de *n* observaciones sobre el comportamiento del sistema, lo cual nos será de utilidad para entender el funcionamiento del mismo –obviamente, nuestro análisis será tanto más preciso cuanto mayor sea el número *n* de experimentos que llevemos a cabo.

Veamos un ejemplo sencillo:

En la imagen inferior se muestra un análisis histórico de 200 días sobre el número de consultas diarias realizadas a un sistema de información empresarial (EIS) residente en un servidor central. La tabla incluye el número de consultas diarias (0 a 5) junto con las frecuencias absolutas (número de días que se producen 0, 1, ..., 5 consultas), las frecuencias relativas (10/200 = 0,05, ...), y las frecuencias relativas acumuladas.

	A	B	C	D	E
1					
2		Consultas EIS	Frec. Abs. (días)	Frec. Relativa	Frec. Relat. Ac.
3		0	10	0,05	0,05
4		1	20	0,10	0,15
5		2	40	0,20	0,35
6		3	60	0,30	0,65
7		4	40	0,20	0,85
8		5	30	0,15	1,00
9		***Total***	***200***	***1,00***	
10					

Podemos interpretar la frecuencia relativa como la probabilidad de que ocurra el suceso asociado, en este caso, la probabilidad de un determinado número de consultas (así, p.e., la probabilidad de que se den 3 consultas en un día sería de 0,30), por lo que la tabla anterior nos proporciona la distribución de probabilidad asociada a una variable aleatoria discreta (la variable aleatoria es el número de consultas al EIS, que sólo puede tomar valores enteros entre 0 y 5).

Supongamos que queremos conocer el número esperado (o medio) de consultas por día. La respuesta a esta pregunta es fácil si recurrimos a la teoría de la probabilidad:

Denotando por X a la variable aleatoria que representa el número diario de consultas al EIS, sabemos que:

$$E[X] = \sum_{i=0}^{5} x_i \cdot P(X = x_i) = 0 \cdot 0{,}05 + 1 \cdot 0{,}10 + \ldots + 5 \cdot 0{,}15 = 2{,}95$$

Por otra parte, también podemos usar simulación de Monte Carlo para estimar el número esperado de consultas diarias (en este caso se ha podido obtener el valor exacto usando teoría de probabilidad, pero ello no siempre será factible). Veamos cómo:

Cuando se conozca la distribución de probabilidad asociada a una variable aleatoria discreta, será posible usar la columna de frecuencias relativas acumuladas para obtener los llamados **intervalos de números aleatorios** asociados a cada suceso. En este caso, los intervalos obtenidos son:

- [0,00 , 0,05) para el suceso 0
- [0,05 , 0,15) para el suceso 1
- [0,15 , 0,35) para el suceso 2
- [0,35 , 0,65) para el suceso 3
- [0,65 , 0,85) para el suceso 4
- [0,85 , 1,00) para el suceso 5

El gráfico siguiente nos muestra cada una de las probabilidades sobre el número de consultas. En él, se aprecia claramente la relación existente entre probabilidad de cada suceso y el área que éste ocupa.



Esto significa que, al generar un número pseudo-aleatorio con el ordenador (proveniente de una distribución uniforme entre 0 y 1), estaremos llevando a cabo un experimento cuyo resultado, obtenido de forma aleatoria y según la distribución de probabilidad anterior, estará asociado a un suceso. Así por ejemplo, si el ordenador nos proporciona el número pseudo-aleatorio 0,2567, podremos suponer que ese día se han producido 2 consultas al EIS.

Asignamos pues la función ALEATORIO a una casilla (la G1 en el caso de la imagen):



G1 = =ALEATORIO()

	A	B	C	D	E	F	G	H
1							0,02755711	
2		Consultas EIS	Frec. Abs. (días)	Frec. Relativa	Frec. Relat. Ac.			

Seleccionando la celda y "arrastrando" con el ratón desde el borde inferior derecho de la misma podemos obtener un listado completo de números pseudo-aleatorios:



92						0,93287364	
93						0,96329721	
94						0,75323996	
95						0,7809585	
96						0,00471791	
97						0,5459221	
98						0,71227576	
99						0,47161286	
100						0,92122832	

A continuación, podemos usar la función **SI** de Excel para asignar un suceso a cada uno de los números pseudo-aleatorios generados (como veremos, otra forma de hacer esta asignación será usando la función **BUSCARV**):

H1 = =SI(G1<E$3;B$3;SI(G1<E$4;B$4;SI(G1<E$5;B$5;SI(G1<E$6;B$6;SI(G1<E$7;B$7;B$8)))))

	A	B	C	D	E	F	G	H
1							0,10772119	1
2		Consultas EIS	Frec. Abs. (días)	Frec. Relativa	Frec. Relat. Ac.		0,1411781	
3		0	10	0,05	0,05		0,72144284	
4		1	20	0,10	0,15		0,50145062	
5		2	40	0,20	0,35		0,9578586	
6		3	60	0,30	0,65		0,34066854	
7		4	40	0,20	0,85		0,91709909	
8		5	30	0,15	1,00		0,04148589	
9		***Total***	***200***	***1,00***			0,64428827	
10							0,76643621	

Repitiendo el proceso de seleccionar y "arrastrar" obtendremos algo similar a:

94							0,20524876	2
95							0,95944028	5
96							0,87429246	5
97							0,63762462	3
98							0,32272825	2
99							0,72548037	4
100							0,15403118	2

Finalmente, usando la función **PROMEDIO** será posible calcular la media de los valores de la columna H:

= =PROMEDIO(H:H)

	C	D	E	F	G	H	I
					0,23259154	2	2,95
S	Frec. Abs. (días)	Frec. Relativa	Frec. Relat. Ac.		0,49286068	3	
	10	0,05	0,05		0,59610971	3	
	20	0,10	0,15		0,60479183	3	

En este caso, hemos obtenido un valor estimado que corresponde exactamente con el valor real anteriormente calculado vía la definición teórica de la media. Sin embargo, debido a la componente aleatoria intrínseca al modelo, normalmente obtendremos valores "cercanos" al valor real, siendo dichos valores diferentes unos de otros (cada simulación proporcionará sus propios resultados). Se puede comprobar este hecho pulsando repetidamente sobre la función F9 (cada vez que se pulsa dicha tecla, Excel genera nuevos valores aleatorios y, por tanto, nuevos valores para la columna H y la casilla I1).

Si en lugar de usar una muestra aleatoria formada por 100 observaciones hubiésemos usado una formada por 10, los valores que obtendríamos al pulsar repetidamente F9 no serían estimaciones tan buenas al valor real. Por el contrario, es de esperar que si hubiésemos usado 1.000 (o mejor aún 10.000) observaciones, los valores que obtendríamos en la casilla I1 estarían todos muy cercanos al valor real.

CASOS PRÁCTICOS CON SOFTWARE

Simulación MC con variables discretas

Veamos un ejemplo algo más complejo del uso de Excel para construir modelos de simulación MC cuando las variables aleatorias sean discretas:

Supongamos que trabajamos en un gran almacén informático, y que nos piden consejo para decidir sobre el número de licencias de un determinado sistema operativo que conviene adquirir – las licencias se suministrarán con los ordenadores que se vendan durante el próximo trimestre, y es lógico pensar que en pocos meses habrá un nuevo sistema operativo en el mercado de características superiores. Cada licencia de sistema operativo le cuesta al almacén un total de 75 Euros, mientras que el precio al que la vende es de 100 Euros. Cuando salga al mercado la nueva versión del sistema operativo, el almacén podrá devolver al distribuidor las licencias sobrantes, obteniendo a cambio un total del 25 Euros por cada una. Basándose en los datos históricos de los últimos meses, los responsables del almacén han sido capaces de determinar la siguiente distribución de probabilidades por lo que a las ventas de licencias del nuevo sistema operativo se refiere:

	A	B	C	D	E	F
1						
2	nº Lic. Vendidas	Probabilidad	Prob. Acum.	Ext. Inf. Intervalo	Ext. Sup. Intervalo	nº Lic. Vendidas
3	100	0,30	0,30	0,00	0,30	100
4	150	0,20	0,50	0,30	0,50	150
5	200	0,30	0,80	0,50	0,80	200
6	250	0,15	0,95	0,80	0,95	250
7	300	0,05	1,00	0,95	1,00	300
8						
9						
10						
11		*Coste x Licencia*	75	Euros		
12		*Ingresos x Lic. Vendida*	100	Euros		
13		*Ingresos x Lic. Devuelta*	25	Euros		
14		*Cantidad Lic. a comprar*	200	Licencias		
15						

Construimos nuestro modelo usando las fórmulas que se muestran en la figura inferior. En la casilla H2 usaremos la función ALEATORIO para generar el valor pseudo-aleatorio que determinará el suceso resultante; en la celda I2 usamos la función BUSCARV para determinar el suceso correspondiente asociado al valor pseudo-aleatorio obtenido –notar que usamos también la función **MIN**, ya que en ningún caso podremos vender más licencias que las disponibles. El resto de fórmulas son bastante claras:

	G	H	I	J	K	L	M	N
1		Num. Aleat.	Lic. Vend.	Lic. Dev.	Coste	Ingresos Venta	Ingresos Dev.	Beneficios
2		=ALEATORIO()	=MIN(BUSCARV(H2;$D$3:$F$7;3);$C$14)	=$C$14-I2	=$C$14*$C$11	=I2*$C$12	=J2*$C$13	=L2+M2-K2
3								

En la imagen anterior se muestra cómo construir el modelo con una observación (iteración). A fin de generar nuevas observaciones, deberemos seleccionar el rango H2:N2 y "arrastrar" hacia abajo (tantas casillas como iteraciones deseemos realizar):

	A	G	H	I	J	K	L	M	N
1			Num. Aleat.	Lic. Vend.	Lic. Dev.	Coste	Ingresos Venta	Ingresos Dev.	Beneficios
2	nº Lic. Vendidas		0,0552506	100	100	15.000	10.000	2.500	-2.500
3	100		0,037224525	100	100	15.000	10.000	2.500	-2.500
4	150		0,793388024	200	0	15.000	20.000	0	5.000
5	200		0,075111272	100	100	15.000	10.000	2.500	-2.500
98			0,705900899	200	0	15.000	20.000	0	5.000
99			0,864060361	200	0	15.000	20.000	0	5.000
100			0,148136027	100	100	15.000	10.000	2.500	-2.500
101			0,483028805	150	50	15.000	15.000	1.250	1.250
102									

Finalmente, es posible estimar el valor esperado de la variable aleatoria que proporciona los beneficios sin más que hallar la media de las 100 observaciones que acabamos de realizar. Asimismo, usaremos las funciones **DESVEST** e **INTERVALO.CONFIANZA** para hallar, respectivamente, la desviación estándar de la muestra obtenida y el intervalo de confianza (a un nivel del 95%) para el valor esperado:

	B	C	D
16			
17	**Variable beneficio**		
18	*Media muestral:*	=PROMEDIO(N:N)	
19	*Desviación estándar:*	=DESVEST(N:N)	
20	*Valor Min.:*	=MIN(N:N)	
21	*Valor Max.:*	=MAX(N:N)	
22	*nivel significación (alpha):*	0,05	
23	*Amplitud Intervalo de confianza:*	=INTERVALO.CONFIANZA(C22;C19;CONTAR(N:N))	
24	*IC para la media a nivel (1-alpha)%:*	=C18-C23	=C18+C23
25			

La apariencia final de nuestro modelo será:



	B	C	D	H	I	J	K	L	M	N
1				Num. Aleat.	Lic. Vend.	Lic. Dev.	Coste	Ingresos Venta	Ingresos Dev.	Beneficios
2	Probabilidad	Prob. Acum.	Ext. Inf. Intervalo	0,188758875	100	100	15.000	10.000	2.500	-2.500
3	0,30	0,30	0,00	0,012241832	100	100	15.000	10.000	2.500	-2.500
4	0,20	0,50	0,30	0,438361797	150	50	15.000	15.000	1.250	1.250
5	0,30	0,80	0,50	0,362533668	150	50	15.000	15.000	1.250	1.250
6	0,15	0,95	0,80	0,654480584	200	0	15.000	20.000	0	5.000
7	0,05	1,00	0,95	0,865143968	200	0	15.000	20.000	0	5.000
8				0,790237576	200	0	15.000	20.000	0	5.000
9				0,134565302	100	100	15.000	10.000	2.500	-2.500
10				0,777159567	200	0	15.000	20.000	0	5.000
11	*Coste x Licencia*	75	Euros	0,695439394	200	0	15.000	20.000	0	5.000
12	*gresos x Lic. Vendida*	100	Euros	0,121616963	100	100	15.000	10.000	2.500	-2.500
13	*resos x Lic. Devuelta*	25	Euros	0,611519942	200	0	15.000	20.000	0	5.000
14	*ntidad Lic. a comprar*	200	Licencias	0,948676046	200	0	15.000	20.000	0	5.000
15				0,158778727	100	100	15.000	10.000	2.500	-2.500
16				0,577209221	200	0	15.000	20.000	0	5.000
17	**Variable beneficio**			0,983086264	200	0	15.000	20.000	0	5.000
18	*Media muestral:*	1.663		0,101529456	100	100	15.000	10.000	2.500	-2.500
19	*Desviación estándar:*	3.409		0,332113889	150	50	15.000	15.000	1.250	1.250
20	*Valor Min.:*	-2.500		0,901835015	200	0	15.000	20.000	0	5.000
21	*Valor Max.:*	5.000		0,133014073	100	100	15.000	10.000	2.500	-2.500
22	*significación (alpha):*	0,05		0,54253693	200	0	15.000	20.000	0	5.000
23	*tervalo de confianza:*	668		0,466520644	150	50	15.000	15.000	1.250	1.250
24	*lia a nivel (1-alpha)%:*	994	2.331	0,910570567	200	0	15.000	20.000	0	5.000
95				0,111809197	100	100	15.000	10.000	2.500	-2.500
96				0,289448378	100	100	15.000	10.000	2.500	-2.500
97				0,883845974	200	0	15.000	20.000	0	5.000
98				0,462640604	150	50	15.000	15.000	1.250	1.250
99				0,804697777	200	0	15.000	20.000	0	5.000
100				0,148195301	100	100	15.000	10.000	2.500	-2.500
101				0,637381618	200	0	15.000	20.000	0	5.000

A partir del modelo anterior es posible también realizar "*what-if*" análisis (análisis de escenarios o preguntas del tipo "¿qué pasaría si cambiamos tal o cual *input*?"). Para ello es suficiente con ir cambiando los valores de las celdas con fondo amarillo o rojo (*inputs* del modelo en este ejemplo). Asimismo, podemos ampliar fácilmente el número de iteraciones (observaciones muestrales) sin más que repetir los procesos de seleccionar y "arrastrar".

En el caso actual, hemos optado por tomar 1.000 iteraciones para cada una de los posibles *inputs* asociados a la cantidad de pedido (estos posibles *inputs* son: 100, 150, 200, 250, y 300). Si se realizase el experimento, se obtendrían unos resultados similares a los que se muestran a continuación (ya que 1.000 es un número ya bastante considerable para este ejemplo):

	Resultados para n = 1.000 iteraciones			
Nº Licencias	Benef. Medio	Desv. Est.	Intervalo Confianza 95%	
100	2.500	0	2.500	2.500
150	2.666	1.701	2.561	2.772
200	1.951	3.305	1.746	2.156
250	261	4.242	-2	524
300	-2.006	4.596	-2.291	-1.721



A partir de los resultados, parece claro que la decisión óptima es hacer un pedido de 150 unidades, ya que con ello se consigue el beneficio máximo.

Generación de números aleatorios provenientes de otras distribuciones

Las últimas versiones de Excel incorporan un *Add-In* llamado **Análisis de datos**. Este complemento proporciona nuevas funcionalidades estadísticas a la hoja de cálculo. Entre ellas, nos interesa destacar la de **Generación de números aleatorios**:





Con esta opción, es posible generar fácilmente observaciones provenientes de diversas distribuciones de variable discreta (Bernoulli, Binomial, Poisson, Frecuencia relativa, y Discreta) o de variable continua (Uniforme y Normal).

Independientemente del complemento Análisis de datos, es posible usar un resultado muy conocido de la teoría estadística, llamado **método de la transformada inversa**, para derivar las fórmulas que permiten obtener valores pseudo-aleatorios provenientes de distribuciones como la Weibull o la Lognormal.

En la tabla siguiente se muestran algunas fórmulas que, implementadas en celdas de Excel, nos permiten obtener valores pseudo-aleatorios de algunas de las distribuciones continuas más usadas:

Distribución	Parámetros	Fórmula Excel
Exponencial	Media = b	= -LN(ALEATORIO())*b
Weibull	Escala = b Forma = a	= b*(-LN(ALEATORIO())^(1/a)
Normal	Media = μ Desv. Estándar = σ	= DISTR.NORM.INV(ALEATORIO(),μ,σ)
Lognormal	Media de Ln(X) = μ Desv. Estándar de Ln(X) = σ	= DISTR.LOG.INV(ALEATORIO(),μ,σ)
Uniforme entre a y b	Extremo inferior = a Extremo superior = b	= a+(b-a)*ALEATORIO()

Añadir, finalmente, que es relativamente sencillo implementar funciones VBA que, haciendo uso del método de la transformada inversa o de otros métodos similares, permitan la generación de valores provenientes de casi cualquier distribución teórica.

Simulación MC con variables continuas

Como hemos comentado, es posible usar las fórmulas anteriores para generar, a partir de la función ALEATORIO(), valores pseudo-aleatorios provenientes de otras distribuciones continuas. En las páginas siguientes, veremos dos ejemplos de modelos que hacen uso de la distribución normal (la distribución estadística más importante y utilizada):

Ejemplo 1: Tiempo de consultas a servidores en paralelo

Supongamos que desde un ordenador cliente se realiza consultas SQL a bases de datos situadas en dos servidores distintos. Nuestro objetivo será estimar el tiempo esperado (tiempo medio) que deberemos esperar para recibir la respuesta de ambos servidores. Dada la complejidad de la consulta que queremos realizar, y basándonos en experiencias anteriores, se calcula que el tiempo necesario para que cada uno de los servidores responda a la misma sigue una distribución normal con los parámetros (media y desviación estándar, en minutos) que se indican a continuación:

	A	B	C	D	E	F
1						
2			Distribución	Media (min)	Desv. Est. (min)	
3		Servidor 1	Normal	20	3,4	
4		Servidor 2	Normal	22	2,6	
5						

Pediremos a Excel que genere valores pseudo-aleatorios provenientes de dichas distribuciones. Asimismo, usaremos la función **MAX** para obtener el tiempo de respuesta (que será el máximo de los tiempos de respuesta de cada servidor), y la función **SI** para determinar qué servidor ha sido el más rápido en responder:

G	H	I	J
Tiempos de respuesta (minutos)			Servidor más rápido
Servidor 1	Servidor 2	Total	
=DISTR.NORM.INV(ALEATORIO();$D$3;$E$3)	=DISTR.NORM.INV(ALEATORIO();$D$4;$E$4)	=MAX(G3:H3)	=SI(G3<H3;1;2)

Usaremos también las funciones **CONTAR** y **CONTAR.SI** para contar el número de iteraciones y el número de veces que un servidor es más rápido que el otro:

	C	D	E
1			
2	**Distribución**	**Media (min)**	**Desv. Est. (min)**
3	Normal	20	3,4
4	Normal	22	2,6
5			
6		Frecuencia	Porcentaje
7	*Total iteraciones:*	=CONTAR(J:J)	=D7/$D$7
8	*S1 + rápido:*	=CONTAR.SI(J:J;1)	=D8/$D$7
9	*S2 + rápido:*	=CONTAR.SI(J:J;2)	=D9/$D$7
10			
11	*Tiempo Medio:*	**=PROMEDIO(I:I)**	minutos
12	*Desv. Est.:*	=DESVEST(I:I)	minutos
13	*95% para Total Esperado:*	**=$D$11-INTERVALO.CONFIANZA(0,05;$D$12;$D$7)**	**=$D$11+INTERVALO.CONFIANZA(0,05;$D$12;$D$7)**
14			

Finalmente, las funciones PROMEDIO, DESVEST, e INTERVALO.CONFIANZA nos servirán para obtener, respectivamente, el tiempo muestral medio (esperado) de respuesta, la desviación estándar de la muestra (observaciones que generaremos), y un intervalo de confianza, a un nivel del 95%, para el tiempo medio (este intervalo nos permitirá saber si nuestra estimación es buena o si, por el contrario, necesitaremos más iteraciones).

Una vez introducidas las fórmulas anteriores, bastará con seleccionar y "arrastrar" hacia abajo el rango de celdas G3:J3, con lo que se generarán nuevas iteraciones. En la imagen siguiente se muestra el resultado obtenido al generar 2.077 iteraciones. Observar que el tiempo medio estimado de respuesta es de 22,9 minutos, y podemos asegurar, con un nivel de confianza del 95%, que dicho tiempo medio estará entre 22,8 y 23,0 minutos.

					Tiempos de respuesta (minutos)			**Servidor**
	Distribución	**Media (min)**	**Desv. Est. (min)**		**Servidor 1**	**Servidor 2**	**Total**	**más rápido**
Servidor 1	Normal	20	3,4		15,0	24,0	24,0	1
Servidor 2	Normal	22	2,6		15,6	18,0	18,0	1
					20,7	22,3	22,3	1
		Frecuencia	Porcentaje		25,3	21,2	25,3	2
	Total iteraciones:	2077	100%		26,2	18,6	26,2	2
	S1 + rápido:	1385	67%		21,5	24,4	24,4	1
	S2 + rápido:	692	33%		16,3	19,5	19,5	1
					17,5	24,7	24,7	1
	Tiempo Medio:	**22,9**	minutos		14,2	23,0	23,0	1
	Desv. Est.:	2,4	minutos		25,9	19,7	25,9	2
	IC 95% para Total Esperado:	**22,8**	**23,0**		21,3	20,5	21,3	2
					23,8	24,4	24,4	1

Finalmente, se observa también que el servidor 1 ha respuesto más rápido que el servidor 2 en el 67% de las iteraciones.

Ejemplo 2: Inversión inicial y flujo de caja

Consideremos ahora un nuevo problema: supongamos que disponemos de un capital inicial de 250 Euros que deseamos invertir en una pequeña empresa. Supondremos también que los flujos de caja -tanto los de entrada como los de salida- son aleatorios, siguiendo éstos una distribución normal.

	A	B	C	D	E
1					
2		***Capital inicial:***	250	Euros	
3					
4	***Flujo de caja Enero***		Distribución	Media	Desv. Est.
5		*Entrante*	Normal	500	125
6		*Saliente*	Normal	400	100
7					

Para el primer mes, el valor esperado del flujo de entrada es de 500 Euros, mientras que el valor esperado para el flujo de salida es de 400 Euros. En meses posteriores, el valor esperado será el valor obtenido para en el mes anterior. Por su parte, las desviaciones estándar valdrán, en todos los casos, un 25% del valor medio (esperado) asociado. En base a lo anterior, podemos construir un modelo como se muestra en las siguientes imágenes:

	G	H	I
1	FLUJO ENERO		
2	Entrante	Saliente	Neto
3	=DISTR.NORM.INV(ALEATORIO();$D$5;$E$5)	=DISTR.NORM.INV(ALEATORIO();$D$6;$E$6)	=$C$2+G3-H3
4			

	J	K	L
1	FLUJO FEBRERO		
2	Entrante	Saliente	Neto
3	=DISTR.NORM.INV(ALEATORIO();G3;0,25*G3)	=DISTR.NORM.INV(ALEATORIO();H3;0,25*H3)	=I3+J3-K3
4			

	M	N	O
1	FLUJO MARZO		
2	Entrante	Saliente	Neto
3	=DISTR.NORM.INV(ALEATORIO();J3;0,25*J3)	=DISTR.NORM.INV(ALEATORIO();K3;0,25*K3)	=L3+M3-N3
4			

	B	C	D
10	*Iteraciones:*	=CONTAR(G:G)	
11	*Flujo Final Max:*	=MAX(O:O)	Euros
12	*Flujo Final Min:*	=MIN(O:O)	Euros
13	*Capital Final Esperado:*	**=PROMEDIO(O:O)**	Euros
14	*IC 95%:*	=$C$13-INTERVALO.CONFIANZA(0,05;DESVEST($O:$O);$C$10)	=$C$13+INTERVA
15			

Seleccionando y "arrastrando" hacia abajo el rango G3:O3, hemos obtenido los siguientes resultados para 5.859 iteraciones:

	A	B	C	D	E	F	G	H	I	J	K
1							FLUJO ENERO			FLUJO	
2		***Capital inicial:***	250	Euros			Entrante	Saliente	Neto	Entrante	Sali
3							291,01	395,62	145,40	430,07	3
4		***Flujo de caja Enero***	Distribución	Media	Desv. Est.		541,36	506,25	285,11	740,18	61
5		*Entrante*	Normal	500	125		646,78	346,67	550,12	376,19	4
6		*Saliente*	Normal	400	100		449,09	526,56	172,53	401,77	3
7							492,15	545,81	196,34	374,14	5
8							418,22	318,54	349,68	454,23	3
9							539,29	283,93	505,36	312,83	1
10		*Iteraciones:*	5859				559,35	348,81	460,54	458,97	2
11		*Flujo Final Max:*	3.211,04	Euros			446,92	440,22	256,70	593,30	41
12		*Flujo Final Min:*	-1.543,65	Euros			367,36	402,95	214,41	326,45	4
13		*Capital Final Esperado:*	543,79	Euros			558,25	422,41	385,84	587,13	4
14		*IC 95%:*	528,47	559,12			541,35	400,85	390,50	719,73	4
15							382,66	367,75	264,91	392,01	1

Observamos que el valor esperado para el capital final es de unos 544 Euros, y que podemos afirmar, con un nivel de confianza del 95%, que dicho valor estará entre 528 y 560 Euros.

BIBLIOGRAFÍA

[1] Seila, A.F. (2001): *Spreadsheet Simulation.* Proceedings of the 2001 Winter Simulation Conference, pp. 74 – 78.

[2] Savage, S.L. (1998): "Insight.xla: Business Analysis Software for Microsoft Excel". Duxbury Press.

[3] Camm, J.D.; Evans, J.R. (1996): "Management Science: Modelling, Analysis and Interpretation". South – Western College Publishing.

[4] Winston, W.L.; Albright, S.C. (1997): "Practical Management Science: Spreadsheet Modeling and Applications". Duxbury Press.

[5] Gedam, S.G.; Beaudet, S.T. (2000): *Monte Carlo Simulation using Excel Spreadsheet for Predicting Reliability of a Complex System.* Proceedings Annual Reliability and Maintainability Symposium.

[6] Evans, J.R. (2000): *Spreadsheets as a Tool for Teaching Simulation*. Informs Transactions On Education Volume 1, Number 1. http://ite.informs.org/Vol1No1/evans/evans.html

[7] Eckstein, J; Riedmueller, S.T. (2002): *YASAI: Yet Another Add-in for Teaching Elementary Monte Carlo Simulation in Excel.* Informs Transactions On Education. Volume 2, Number 2. http://ite.informs.org/vol2no2/EcksteinRiedmueller/

[8] Judge, G. (1999): *Simple Monte Carlo studies on a spreadsheet.* Computers in Higher Education Economics Review (CHEER). Volume 13, Issue 2. http://www.economics.ltsn.ac.uk/cheer/ch13_2/ch13_2p12.htm

[9] Hwarng, H.B. (2001): *A Modern Simulation Course for Business Students*. Interfaces, 31:3, Part 1 of 2, May-June 2001, pp. 66-75.

[10] Nance, R.E., Sargent, R.G. (2002): *Perspectives on the Evolution of Simulation*. Operations Research, Vol. 50, No. 1, January-February 2002, pp. 161-172.

ENLACES

[W1] http://www.geocities.com/CollegePark/Quad/2435/index.html
Breve historia de los orígenes del método Monte Carlo.

[W2] http://www.palisade.com/
Página web de @Risk

[W3] http://www.crystalball.com/
Página web de Crystal Ball

[W4] http://www.kellogg.nwu.edu/faculty/myerson/ftp/addins.htm
Página web de SimTools.xla

[W5] http://analycorp.com/stan/
Página web de Insight.xla

[W6] http://www.projectware.com.au/tutorials/Tu08.pdf
Breve artículo sobre simulación MC con Excel

[W7] http://www.barringer1.com/MC.htm
Página web donde se comentan algunas aplicaciones de Excel y simulación MC para el estudio de la fiabilidad de sistemas

[W8] http://random.mat.sbg.ac.at/links/index.html
Página web de la WWW Virtual Library dedicada a números aleatorios y simulación MC

[W9] http://csep1.phy.ornl.gov/mc/mc.html
Libro electrónico sobre simulación MC

[W10] http://www.geocities.com/WallStreet/9245/vba.htm
Página web donde se muestran algunos ejemplos de simulación MC con Excel y VBA

[W11] http://www.csun.edu/~vcmgt0j3/Ch12Notes.pdf
Apuntes sobre simulación MC con Excel

[W12] http://www.wabash.edu/depart/economic/EconExcel/home.htm
Modelos económicos y econométricos con Excel (algunos usan simulación MC)